



Transformations linéaires et géométrie dans le plan

© Pierre Lantagne (Septembre 2015)

Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2001. L'objectif principal de cette feuille Maple est de présenter les matrices comme objets pouvant agir directement sur les vecteurs. Plus précisément, c'est pour montrer comment la multiplication matricielle peut être utilisée pour opérer des transformations géométriques dans le plan.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2015.2

Initialisation

```
> restart;  
> with(plots,display,setoptions):  
setoptions(axesfont=[TIMES,ROMAN,8],size=[200,200]); # Pour le pdf
```

Transformations linéaires

Une transformation linéaire est une application $T: U \rightarrow V$ entre deux espaces vectoriels sur un même corps K , vérifiant pour tous $\mathbf{u}, \mathbf{v} \in V$ et pour tout $\alpha \in K$,

$$-T(\mathbf{u}+\mathbf{v}) = T(\mathbf{u})+T(\mathbf{v})$$

$$-T(\alpha\mathbf{v}) = \alpha T(\mathbf{v})$$

Pour simplifier, nous dirons que T est une transformation linéaire si l'application T préserve l'addition et la multiplication par un scalaire.

Par exemple, montrons que l'application $T: \mathbb{R}^2 \rightarrow \mathbb{R}^3$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix}$ est une transformation linéaire.

Créons ponctuellement une macro-commande T opérant sur un vecteur $\begin{bmatrix} x \\ y \end{bmatrix}$ de dimension 2 dont le résultat

$$\text{sera } T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix}.$$

```
> T:=proc(v::Vector)  
local w;  
w:=convert(v,list);
```

```

    if nops(w)<>2 then ERROR("L'argument de T doit être un vecteur de dimension
2") fi;
    convert([w[1]+w[2],w[1]-w[2],w[2]],Vector)
end:

```

Montrons que la macro-commande T est bien la transposition en Maple de l'application T.

Soit $u \in \mathbb{R}^2$ quelconque.

```
> u:=<x,y>;
```

$$u := \begin{bmatrix} x \\ y \end{bmatrix} \quad (2.1)$$

Obtenons le résultat $T(u)$.

```
> 'T(u)'='T'(u);
` `='T(u);
```

$$\begin{aligned} T(u) &= T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) \\ &= \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix} \end{aligned} \quad (2.2)$$

```
> u:='u':
```

Vérifions maintenant si T préserve l'addition et la multiplication par un scalaire.

Soit deux vecteurs quelconques u et v de $\mathbb{R}^2$ et soit α un nombre réel quelconque.

```
> Vecteur_u:=Vector(2,symbol=u);
Vecteur_v:=Vector(2,symbol=v);
```

$$\begin{aligned} \text{Vecteur}_u &:= \begin{bmatrix} u_1 \\ u_2 \end{bmatrix} \\ \text{Vecteur}_v &:= \begin{bmatrix} v_1 \\ v_2 \end{bmatrix} \end{aligned} \quad (2.3)$$

```
> Vecteur_u:=<u[1],u[2]>;
Vecteur_v:=<v[1],v[2]>;
```

$$\begin{aligned} \text{Vecteur}_u &:= \begin{bmatrix} u_1 \\ u_2 \end{bmatrix} \\ \text{Vecteur}_v &:= \begin{bmatrix} v_1 \\ v_2 \end{bmatrix} \end{aligned} \quad (2.4)$$

Développons d'une part l'image $T(u+v)$:

```
> 'T(Vecteur_u+Vecteur_v)'='T(<u[1],u[2]>+<v[1],v[2]>)' ;
` `='T'(Vecteur_u+Vecteur_v);
```

```
` `=T(Vecteur_u+Vecteur_v) ;
```

$$T(\text{Vecteur}_u + \text{Vecteur}_v) = T(\langle u_1, u_2 \rangle + \langle v_1, v_2 \rangle)$$

$$= T\left(\begin{bmatrix} u_1 + v_1 \\ u_2 + v_2 \end{bmatrix}\right)$$

$$= \begin{bmatrix} u_1 + v_1 + u_2 + v_2 \\ u_1 + v_1 - u_2 - v_2 \\ u_2 + v_2 \end{bmatrix}$$

(2.5)

D'autre part, développons la somme des images $T(u) + T(v)$:

```
> 'T(Vecteur_u)+T(Vecteur_v)'='T(<u[1],u[2]>)+'T(<v[1],v[2]>)' ;
```

```
` `='T'('(<u[1],u[2]>)+(<v[1],v[2]>)' ) ;
```

```
` `='T'('(<u[1]+u[2],v[1]+v[2]>)' ) ;
```

```
` `=T(Vecteur_u)+T(Vecteur_v) ;
```

$$T(\text{Vecteur}_u) + T(\text{Vecteur}_v) = T(\langle u_1, u_2 \rangle) + T(\langle v_1, v_2 \rangle)$$

$$= T(\langle u_1, u_2 \rangle + \langle v_1, v_2 \rangle)$$

$$= T(\langle u_1 + u_2, v_1 + v_2 \rangle)$$

$$= \begin{bmatrix} u_1 + v_1 + u_2 + v_2 \\ u_1 + v_1 - u_2 - v_2 \\ u_2 + v_2 \end{bmatrix}$$

(2.6)

Deux quantités égales à une même troisième sont égales entre elles. Donc $T(u + v) = T(u) + T(v)$. Ce qui montre que T préserve l'addition.

Montrons ensuite que T préserve également la multiplication par un scalaire. Développons d'une part l'image $T(\alpha v)$:

```
> 'T(alpha*Vecteur_v)'='T'(alpha.'<v[1],v[2]>') ;
```

```
` `='T'(<alpha*v[1],alpha*v[2]>) ;
```

```
` `=T(<alpha*v[1],alpha*v[2]>) ;
```

```
` `=map(factor,rhs(%)) ;
```

$$T(\alpha \text{Vecteur}_v) = T(\alpha \cdot \langle v_1, v_2 \rangle)$$

$$= T\left(\begin{bmatrix} \alpha v_1 \\ \alpha v_2 \end{bmatrix}\right)$$

$$= \begin{bmatrix} \alpha v_1 + \alpha v_2 \\ \alpha v_1 - \alpha v_2 \\ \alpha v_2 \end{bmatrix}$$

(2.7)

$$= \begin{bmatrix} \alpha(v_1 + v_2) \\ \alpha(v_1 - v_2) \\ \alpha v_2 \end{bmatrix} \quad (2.7)$$

D'autre part multiplions par α l'image $T(v)$:

```
> alpha*T(Vecteur_v) '=alpha*T' ('<v[1],v[2]>');
` `=alpha * 'T' (<v[1],v[2]>);
` `=alpha . T(<v[1],v[2]>);
` `=alpha * T(<v[1],v[2]>);
```

$$\begin{aligned} \alpha T(\text{Vecteur}_v) &= \alpha T(\langle v_1, v_2 \rangle) \\ &= \alpha T\left(\begin{bmatrix} v_1 \\ v_2 \end{bmatrix}\right) \\ &= \alpha \cdot \begin{bmatrix} v_1 + v_2 \\ v_1 - v_2 \\ v_2 \end{bmatrix} \\ &= \begin{bmatrix} \alpha(v_1 + v_2) \\ \alpha(v_1 - v_2) \\ \alpha v_2 \end{bmatrix} \end{aligned} \quad (2.8)$$

Deux quantités égales à une même troisième sont égales entre elles. Donc $T(\alpha v) = \alpha T(v)$. Ce qui montre que T préserve la multiplication par un scalaire.

Nous pouvons donc conclure que T est bien une transformation linéaire de $T: \mathbb{R}^2 \rightarrow \mathbb{R}^3$.

Matrice associée à une transformation linéaire

Soit $T: V \rightarrow W$ une transformation linéaire de l'espace vectoriel V vers l'espace vectoriel W . Lorsqu'il s'agit d'associer une matrice à une transformation linéaire, nous devons faire cette association par rapport à des bases ordonnées de chaque espace vectoriel impliqué au moment de la transformation linéaire. Dans le cas où l'on recherche la matrice associée par rapport aux bases ordonnées de V et de W , les vecteurs colonnes de la matrice associée sont les scalaires de la combinaison linéaire des vecteurs images dans la base ordonnée de W . Autrement dit, chaque colonne de la matrice associée est, dans l'ordre, les scalaires de la combinaison linéaire de $T(v)$ ($T(v) \in W$) dans la base ordonnée de W . Si $\dim(V) = n$ et $\dim(W) = m$, il découle que le format de la matrice associée est alors $m \times n$.

Étude d'un cas particulier

Soit V et W deux espaces vectoriels sur le corps K .

Soit $T: V \rightarrow W$, une transformation linéaire de V vers W .

Supposons que $\dim(V) = 3$ et $\dim(W) = 2$. Choisissons une base ordonnée pour V , disons $\{v_1, v_2, v_3\}$ et une base ordonnée pour W , disons $\{w_1, w_2\}$. Sachant que les vecteurs $T(v_1)$, $T(v_2)$ et $T(v_3)$ de W s'écrivent de façon unique comme combinaison linéaire des vecteurs de la base ordonnée $\{w_1, w_2\}$ de W , nous pouvons écrire:

$$-T(v_1) = a_1 w_1 + a_2 w_2$$

$$-T(v_2) = b_1 w_1 + b_2 w_2$$

$$-T(v_3) = c_1 w_1 + c_2 w_2$$

où les scalaires a_1, a_2, b_1, b_2, c_1 et c_2 sont uniques.

Avec ces scalaires, formons alors la matrice associée $\begin{bmatrix} a_1 & b_1 & c_1 \\ a_2 & b_2 & c_2 \end{bmatrix}$ à la transformation T . Nous avons donc formé une matrice de format 2×3 .

Reprenons la transformation linéaire T de la section précédente, c'est-à-dire $T: \mathbb{R}^2 \rightarrow \mathbb{R}^3$ définie par

$$T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix} \text{ et soit les bases canoniques ordonnées respectives de } \mathbb{R}^2 \text{ et de } \mathbb{R}^3:$$

$$- \text{base naturelle de } \mathbb{R}^2: \{e_1, e_2\} = \left\{ \begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix} \right\}$$

$$- \text{base naturelle de } \mathbb{R}^3: \{e_1, e_2, e_3\} = \left\{ \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \right\}.$$

Recherchons, par rapport à ces bases ordonnées, la matrice associée à la transformation linéaire T . Pour cela, nous devons rechercher les images par T des vecteurs de la base naturelle de $\mathbb{R}^2$ et les exprimer comme combinaison linéaire des vecteurs de la base naturelle de $\mathbb{R}^3$.

```
> e[1]:=Vector(2, [1,0]);
'T'(e[1])=T(e[1]);
```

$$e_1 := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$T\left(\begin{bmatrix} 1 \\ 0 \end{bmatrix}\right) = \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}$$

(3.1)

$$\text{Il est évident ici que le vecteur } T(e_1) = \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} = 1 \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + 1 \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} + 0 \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}.$$

```
> e[2]:=Vector(2, [0,1]);
'T'(e[2])=T(e[2]);
```

$$e_2 := \begin{bmatrix} 0 \\ 1 \end{bmatrix}$$

$$T\left(\begin{bmatrix} 0 \\ 1 \end{bmatrix}\right) = \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix} \quad (3.2)$$

Il est tout aussi évident que le vecteur $T(e_2) = \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix} = 1 \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} - 1 \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} + 1 \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$.

Reste donc à disposer les scalaires de chaque combinaison linéaire sur deux colonnes.

```
> A:=matrix([
    [1, 1],
    [1,-1],
    [0,1] ]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 1 & -1 \\ 0 & 1 \end{bmatrix} \quad (3.3)$$

Le choix des bases naturelles nous a beaucoup simplifié le calcul dans la recherche de la matrice associée à la transformation linéaire T . De plus, il est facile de comprendre qu'avec la base naturelle $\{e_1, e_2, e_3\}$ de $\mathbb{R}^3$, si on change l'ordre des vecteurs de cette base, la matrice associée sera différente.

Écriture matricielle d'une transformation linéaire

La matrice A dans l'équation matricielle $A \mathbf{X} = \mathbf{B}$ peut être perçue comme un objet mathématique qui «transforme» un vecteur colonne $\mathbf{X}$ en un nouveau vecteur colonne $\mathbf{B}$. Soit V un espace vectoriel de dimension n et soit W un espace vectoriel de dimension m . Si $A_{m \times n}$ est la matrice associée à une transformation linéaire $T: V \rightarrow W$ et si $\mathbf{v}$ est un vecteur colonne de V alors $A \mathbf{v}$ sera l'image du vecteur $\mathbf{v}$ par la transformation linéaire T .

$$A \mathbf{v} = T(\mathbf{v})$$

$A \mathbf{v}$ est donc un vecteur colonne de W .

Exemple

Considérons de nouveau la transformation $T: \mathbb{R}^2 \rightarrow \mathbb{R}^3$ défini par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix}$.

On a montré que T est bien une transformation linéaire et que la matrice $A = \begin{bmatrix} 1 & 1 \\ 1 & -1 \\ 0 & 1 \end{bmatrix}$ est bien la matrice

associée à la transformation T par rapport aux bases naturelles de $\mathbb{R}^2$ et de $\mathbb{R}^3$.

D'une part, obtenons l'image du vecteur $\mathbf{v} = \begin{bmatrix} 5 \\ -6 \end{bmatrix}$ par la transformation T .

```
> v:=<5,-6>:
> 'T'(v)=T(v);
```

$$T\left(\begin{bmatrix} 5 \\ -6 \end{bmatrix}\right) = \begin{bmatrix} -1 \\ 11 \\ -6 \end{bmatrix} \quad (4.1)$$

D'autre part, obtenons le produit matriciel $A \mathbf{v}$.

```
> A*v=eval(A) * v;
\ \=A.v;
```

$$\begin{aligned} A \begin{bmatrix} 5 \\ -6 \end{bmatrix} &= \begin{bmatrix} 1 & 1 \\ 1 & -1 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} 5 \\ -6 \end{bmatrix} \\ &= \begin{bmatrix} -1 \\ 11 \\ -6 \end{bmatrix} \end{aligned} \quad (4.2)$$

Plus généralement, soit $\mathbf{v} = \begin{bmatrix} x \\ y \end{bmatrix}$ un vecteur quelconque de $\mathbb{R}^2$ et obtenons le produit $A \mathbf{v}$ où, cette fois, $\mathbf{v} =$

$$\begin{bmatrix} x \\ y \end{bmatrix}.$$

```
> v:=<x,y>
```

$$\mathbf{v} := \begin{bmatrix} x \\ y \end{bmatrix} \quad (4.3)$$

```
> A*v=eval(A) * v;
\ \=A.v;
```

$$\begin{aligned} A \begin{bmatrix} x \\ y \end{bmatrix} &= \begin{bmatrix} 1 & 1 \\ 1 & -1 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} \\ &= \begin{bmatrix} x+y \\ x-y \\ y \end{bmatrix} \end{aligned} \quad (4.4)$$

Ce dernier résultat montre bien que $A \mathbf{v} = T(\mathbf{v})$.

Il est facile d'accepter que toute matrice réelle A de format $m \times n$ définit une transformation linéaire de $\mathbb{R}^n$ vers $\mathbb{R}^m$. On peut montrer aussi que toute transformation linéaire $T: \mathbb{R}^n \rightarrow \mathbb{R}^m$ possède une unique matrice associée A telle que $A \mathbf{v} = T(\mathbf{v})$, pour tout $\mathbf{v} \in \mathbb{R}^n$.

Principales transformations élémentaires de $\mathbb{R}^2$ vers $\mathbb{R}^2$

Famille des réflexions

Réflexion par rapport à l'axe des x

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} x \\ -y \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

Réflexion par rapport à l'axe des y

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} -x \\ y \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}$$

Réflexion par rapport à la droite $y = x$

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} y \\ x \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

Réflexion par rapport à la droite $y = -x$

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} -y \\ -x \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}$$

Réflexion par rapport à la droite $y = \tan(\theta) x$

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} \cos(2\theta) x + \sin(2\theta) y \\ \sin(2\theta) x - \cos(2\theta) y \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} \cos(2\theta) & \sin(2\theta) \\ \sin(2\theta) & -\cos(2\theta) \end{bmatrix}$$

Réflexion par rapport à l'origine

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} -x \\ -y \end{bmatrix}$$

– Matrice standard associée $\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

Famille des étirements

si $k > 1$, l'étirement est une dilatation

si $0 < k < 1$, l'étirement est une compression

Étirement suivant l'axe des x de facteur k

– $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} kx \\ y \end{bmatrix}$ où $k > 0$

– Matrice standard associée $\begin{bmatrix} k & 0 \\ 0 & 1 \end{bmatrix}$

Étirement suivant l'axe des y de facteur k

– $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x \\ ky \end{bmatrix}$ où $k > 0$

– Matrice standard associée $\begin{bmatrix} 1 & 0 \\ 0 & k \end{bmatrix}$

Étirement selon les deux axes de facteurs k_1 et k_2

– $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} k_1 x \\ k_2 y \end{bmatrix}$

– Matrice standard associée $\begin{bmatrix} k_1 & 0 \\ 0 & k_2 \end{bmatrix}$

Lorsque $k_1 = k_2 = k > 0$, l'étirement est une homothétie de centre $(0,0)$ et de rapport k

Famille des cisaillements

Cisaillement horizontal de facteur k

– $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} kx \\ kx + y \end{bmatrix}$ où $k \neq 0$

– Matrice standard associée $\begin{bmatrix} 1 & k \\ 0 & 1 \end{bmatrix}$

Cisaillement vertical de facteur k

– $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x + ky \\ y \end{bmatrix}$ où $k \neq 0$

– Matrice standard associée $\begin{bmatrix} 1 & 0 \\ k & 1 \end{bmatrix}$

Famille des projections perpendiculaires

Projection perpendiculaire sur l'axe des x

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} x \\ 0 \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} 1 & 0 \\ 0 & 0 \end{bmatrix}$$

Projection perpendiculaire sur l'axe des y

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} 0 \\ y \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

Rotation d'un angle θ autour de l'origine

Si angle $\theta > 0$, le sens de la rotation est le sens anti-horaire

Si angle $\theta < 0$, le sens de la rotation est le sens horaire

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} x \cos(\theta) + y \sin(\theta) \\ -x \sin(\theta) + y \cos(\theta) \end{bmatrix}$$

$$- \text{Matrice standard associée } \begin{bmatrix} \cos(\theta) & -\sin(\theta) \\ \sin(\theta) & \cos(\theta) \end{bmatrix}$$

Rotation-étirement

Bien que la transformation linéaire rotation-étirement ne soit pas une transformation élémentaire, elle est quand même présentée. Cette transformation est en fait la composée de deux transformations élémentaires: une rotation suivie d'une homothétie.

$$-T: \mathbb{R}^2 \rightarrow \mathbb{R}^2 \text{ définie par } T \left(\begin{bmatrix} x \\ y \end{bmatrix} \right) = \begin{bmatrix} ax + by \\ -bx + ay \end{bmatrix} \text{ où } a, b \in \mathbb{R}$$

$$- \text{Matrice standard associée } \begin{bmatrix} a & -b \\ b & a \end{bmatrix}$$

Remarque: La matrice précédente a la forme de la matrice associée à une rotation où l'angle θ de rotation est $\theta = \arctan\left(\frac{b}{a}\right)$. D'autre part, l'étirement est une homothétie de centre $(0,0)$ et de rapport $k = \sqrt{a^2 + b^2}$.

Quelques illustrations graphiques

```
> with(LinearAlgebra, SubMatrix);  
with(plottools, polygon);  
with(plots, setoptions, display);
```

```

setoptions (scaling=constrained, font=[TIMES,ROMAN,10]) ;
[SubMatrix]
[polygon]
[setoptions, display]

```

(6.1)


Réflexion par rapport à l'axe des x

Illustrons graphiquement quelques transformations dans le plan du polygone convexe suivant.

```

> Sommets_O:=[ [0,0], [4,0], [4,2], [2,4], [0,2], [0,0] ] :
> Polygone_O:=polygon(Sommets_O,color=turquoise) :
display(Polygone_O,view=[-2..6,-2..4]) ;

```



Soit la transformation linéaire $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x \\ -y \end{bmatrix}$. On appelle cette transformation *réflexion par rapport à l'axe des x*. La matrice associée à cette transformation (matrice associée par rapport à la base naturelle de $\mathbb{R}^2$) est la matrice $A = \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$. On laissera au soin du lecteur de vérifier si T est bien une transformation linéaire et si la matrice A est bien la matrice standard associée.

Pour illustrer efficacement la transformation sur les segments composant le polygone précédent, créons la matrice des vecteurs colonnes dont les composantes, dans la base naturelle, correspondent aux coordonnées cartésiennes de chaque sommet du polygone.

```

> M_Sommets:=Matrix([
    [0,4,4,2,0,0],
    [0,0,2,4,2,0] ] ) ;

```

$$M_Sommets := \begin{bmatrix} 0 & 4 & 4 & 2 & 0 & 0 \\ 0 & 0 & 2 & 4 & 2 & 0 \end{bmatrix}$$

(6.1.1)

Pour obtenir la matrice précédente, nous aurions pu automatiser notre tâche à partir des coordonnées des points de la liste `Sommets_O`.

```
> M_Sommets:=Matrix([
    [seq(op(1,Sommets_O[i]),i=1..nops(Sommets_O))],
    [seq(op(2,Sommets_O[i]),i=1..nops(Sommets_O))]]);
```

$$M_Sommets := \begin{bmatrix} 0 & 4 & 4 & 2 & 0 & 0 \\ 0 & 0 & 2 & 4 & 2 & 0 \end{bmatrix} \quad (6.1.2)$$

Obtenons maintenant les images des vecteurs colonnes de la matrice $M_Sommets$ en calculant le produit matriciel $A M_Sommets$.

```
> A:=Matrix([
    [1, 0],
    [0,-1]]);
T_Sommets:=A.M_Sommets;
```

$$A := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

$$T_Sommets := \begin{bmatrix} 0 & 4 & 4 & 2 & 0 & 0 \\ 0 & 0 & -2 & -4 & -2 & 0 \end{bmatrix} \quad (6.1.3)$$

Les composantes, dans la base naturelle, de chaque vecteur colonne de la matrice produit $T_Sommets$ correspondent aux coordonnées cartésiennes transformées de chaque sommet du polygone.


Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k),list),k=1..nops
    (Sommets_O))];
```

$$Sommets_T := [[0, 0], [4, 0], [4, -2], [2, -4], [0, -2], [0, 0]] \quad (6.1.4)$$

Traçons le polygone transformé avec la couleur « navy » et superposons-le dans un même graphique avec le polygone original de couleur « orange ».

```
> Polygone_T:=polygon(Sommets_T,color=pink):
display([Polygone_O,Polygone_T],view=[-5..5,-5..5]);
```



Homothétie de centre (0,0) et de rapport 2

Soit l'homothétie centre (0,0) et de rapport 2 $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} 2x \\ 2y \end{bmatrix}$ dont la

matrice standard associée $A = \begin{bmatrix} 2 & 0 \\ 0 & 2 \end{bmatrix}$.

Obtenons maintenant les images des sommets du polygone par la transformation T en calculant le produit matriciel $AM_Sommets$.

```
> A:=Matrix([
    [2,0],
    [0,2] ]);
T_Sommets:=A.M_Sommets;
```

$$A := \begin{bmatrix} 2 & 0 \\ 0 & 2 \end{bmatrix}$$

$$T_Sommets := \begin{bmatrix} 0 & 8 & 8 & 4 & 0 & 0 \\ 0 & 0 & 4 & 8 & 4 & 0 \end{bmatrix} \quad (6.2.1)$$


Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k),list),k=1..nops
    (Sommets_O))];
Sommets_T:= [[0,0], [8,0], [8,4], [4,8], [0,4], [0,0]]
```

(6.2.2)

Traçons le polygone transformé avec la couleur « navy » et superposons-le dans un même graphique avec le polygone original de couleur « orange ».

```
> Polygone_T:=polygon(Sommets_T,color=pink):
display([Polygone_O,Polygone_T],view=[-5..10,-5..10]);
```



Cisaillement horizontal de facteur 1.25

Soit le cisaillement horizontal suivant de facteur $\frac{5}{4}$ $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x \\ \frac{5}{4}x \end{bmatrix}$

dont la matrice standard associée $A = \begin{bmatrix} 1 & \frac{5}{4} \\ 0 & 1 \end{bmatrix}$.

Obtenons maintenant les images des sommets du polygone par la transformation T en calculant le produit matriciel $AM_Sommets$.

```
> A:=Matrix([
    [1,5/4],
    [0, 1 ]]);
T_Sommets:=A.M_Sommets;
```

$$A := \begin{bmatrix} 1 & \frac{5}{4} \\ 0 & 1 \end{bmatrix}$$

$$T_Sommets := \begin{bmatrix} 0 & 4 & \frac{13}{2} & 7 & \frac{5}{2} & 0 \\ 0 & 0 & 2 & 4 & 2 & 0 \end{bmatrix} \quad (6.3.1)$$


Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k),list),k=1..nops
(Sommets_O))];
```

$$Sommets_T := \left[[0, 0], [4, 0], \left[\frac{13}{2}, 2 \right], [7, 4], \left[\frac{5}{2}, 2 \right], [0, 0] \right] \quad (6.3.2)$$

Traçons le polygone transformé avec la couleur « navy » et superposons-le dans un même graphique avec le polygone original de couleur « orange ».

```
> Polygone_T:=polygon(Sommets_T,color=pink):
display([Polygone_T,Polygone_O],view=[-2..8,-2..4],scaling=constrained);
```



Rotation d'un angle $\theta = \frac{2\pi}{3}$

Soit la rotation autour de l'origine $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par

$$T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x \cos\left(\frac{2\pi}{3}\right) + y \sin\left(\frac{2\pi}{3}\right) \\ -x \sin\left(\frac{2\pi}{3}\right) + y \cos\left(\frac{2\pi}{3}\right) \end{bmatrix} \quad \text{dont la matrice standard associée est la matrice } A =$$

$$\begin{bmatrix} \cos\left(\frac{2\pi}{3}\right) & -\sin\left(\frac{2\pi}{3}\right) \\ \sin\left(\frac{2\pi}{3}\right) & \cos\left(\frac{2\pi}{3}\right) \end{bmatrix}.$$

Obtenons maintenant les images des sommets du polygone par la transformation T en calculant le produit matriciel $A M_Sommets$.

```
> A:=Matrix([
    [cos(2*Pi/3), -sin(2*Pi/3)],
    [sin(2*Pi/3), cos(2*Pi/3)] ]);
T_Sommets:=A.M_Sommets;
```

$$A := \begin{bmatrix} -\frac{1}{2} & -\frac{1}{2}\sqrt{3} \\ \frac{1}{2}\sqrt{3} & -\frac{1}{2} \end{bmatrix}$$


$$T_Sommets := \begin{bmatrix} 0 & -2 & -2-\sqrt{3} & -1-2\sqrt{3} & -\sqrt{3} & 0 \\ 0 & 2\sqrt{3} & 2\sqrt{3}-1 & \sqrt{3}-2 & -1 & 0 \end{bmatrix} \quad (6.4.1)$$

Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k), list), k=1..nops
    (Sommets_O))];
Sommets_T:= [[0,0], [-2, 2*sqrt(3)], [-2-sqrt(3), 2*sqrt(3)-1], [-1-2*sqrt(3), sqrt(3)-2], [-sqrt(3),
    -1], [0,0]] (6.4.2)
```

Traçons le polygone transformé avec la couleur « navy » et superposons-le dans un même graphique avec le polygone original de couleur « orange ».

```
> Polygone_T:=polygon(Sommets_T, color=pink);
display([Polygone_O, Polygone_T], view=[-6..6, -6..6]);
```



Pour s'amuser, animons cette dernière transformation mais avec un angle de rotation de 2π radians. Commençons par définir l'angle de rotation ainsi que le nombre d'images de l'animation.

```
> theta:=2*Pi;
   Nombre_images:=36;

                                 $\theta := 2\pi$ 
                                Nombre_images := 36 (6.4.3)
```

Calculons les différentes valeurs d'angles qui serviront à créer les images intermédiaires de l'animation de cette rotation d'angle θ .

```
> Divisions:=[seq(theta-(theta/Nombre_images)*(k-1),k=1..Nombre_images+1)]:
   Pas:=[seq(Divisions[nops(Divisions)-k],k=0..nops(Divisions)-1)]:
Pas := [0, 1/18 pi, 1/9 pi, 1/6 pi, 2/9 pi, 5/18 pi, 1/3 pi, 7/18 pi, 4/9 pi, 1/2 pi, 5/9 pi, 11/18 pi, 2/3 pi, 13/18 pi, 7/9 pi,
5/6 pi, 8/9 pi, 17/18 pi, pi, 19/18 pi, 10/9 pi, 7/6 pi, 11/9 pi, 23/18 pi, 4/3 pi, 25/18 pi, 13/9 pi, 3/2 pi, 14/9 pi,
29/18 pi, 5/3 pi, 31/18 pi, 16/9 pi, 11/6 pi, 17/9 pi, 35/18 pi, 2pi] (6.4.4)
```

Générons ensuite chaque matrice de rotation servant au calcul du tracé de chaque image intermédiaire.


```
> A_animation:=seq(Matrix([
                                cos(theta), -sin(theta)],
                                [sin(theta), cos(theta)]],theta=Pas):
```

Créons la séquence du tracé de chaque image intermédiaire de l'animation.

```
> Polygone_A:=seq(polygon([seq(convert(SubMatrix(M.M_Sommets, 1..2, k),
list),k=1..nops(Sommets_0))],color=pink),M=A_animation):
> Animation:=seq(display([Polygone_O,Polygone_A[k],Polygone_O]),k=1..nops
(Pas)):
```

Reste à faire l'affichage de l'animation.

```
> display(Animation,view=[-6..6,-5..5],insequence=true);
```

Cliquer sur le graphique précédent pour le sélectionner. Cela fera apparaître la barre contextuelle des boutons dédiés aux animations. Cliquer ensuite sur le bouton de départ pour lancer l'animation.

Vous pouvez modifier la valeur de l'angle θ et exécuter à nouveau les requêtes précédentes pour créer de nouvelles animations. Essayez donc avec $\theta = \frac{5\pi}{4}$ et avec $\theta = -\frac{7\pi}{2}$.

Rotation-étirement.

Soit la rotation-étirement d'angle $\theta = \arctan(1)$ et de facteur $k = \sqrt{2^2 + 2^2} = 2\sqrt{2} : \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par

$$T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} 2x + 2y \\ -2x + 2y \end{bmatrix} \text{ dont la matrice standard associée } A = \begin{bmatrix} 2 & -2 \\ 2 & 2 \end{bmatrix}.$$

Obtenons maintenant les images des sommets du polygone original de couleur « orange » par la transformation T en calculant le produit matriciel $A M_Sommets$.

```
> A:=Matrix([
    [2,-2],
    [2, 2] ]);
T_Sommets:=A.M_Sommets;
```

$$A := \begin{bmatrix} 2 & -2 \\ 2 & 2 \end{bmatrix}$$

$$T_Sommets := \begin{bmatrix} 0 & 8 & 4 & -4 & -4 & 0 \\ 0 & 8 & 12 & 12 & 4 & 0 \end{bmatrix} \quad (6.5.1)$$

Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.


```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k),list),k=1..nops
(Sommets_O))];
```

$$Sommets_T := [[0, 0], [8, 8], [4, 12], [-4, 12], [-4, 4], [0, 0]] \quad (6.5.2)$$

Traçons le polygone transformé avec la couleur « navy » et superposons-le dans un même graphique avec

le polygone original de couleur « orange ».

```
> Polygone_T:=polygon(Sommets_T,color=pink):  
display([Polygone_O,Polygone_T],view=[-8..10,-2..14]);
```



Composition de transformations linéaires: un aperçu

Soit $S: \mathbb{R}^p \rightarrow \mathbb{R}^n$ et $T: \mathbb{R}^p \rightarrow \mathbb{R}^m$ deux transformations linéaires. Il est facile de montrer que la composition des deux transformations linéaires définie par $T(S(v))$ est aussi une transformation linéaire mais de $\mathbb{R}^p \rightarrow \mathbb{R}^m$.

De plus, si A_S et A_T sont les matrices associées aux transformations linéaires S et T , alors le produit matriciel $A_S A_T$ est la matrice associée à la composition des deux transformations linéaires S et T .

Par exemple, si on réalise une réflexion par rapport à l'axe des y après un cisaillement horizontal de facteur 1,

25, la matrice associée à la transformation finale sera le produit $A = \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & \frac{5}{4} \\ 0 & 1 \end{bmatrix}$.

```
> A:='A':  
> A[S]:=<<-1|0>,<0|1>>;  
A[T]:=<<1|5/4>,<0|1>>;
```

$$A_S := \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$A_T := \begin{bmatrix} 1 & \frac{5}{4} \\ 0 & 1 \end{bmatrix}$$

(7.1)

```
> A=evalm(A[S])*A[T];  
> A=A[S].A[T];
```

$$A = \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & \frac{5}{4} \\ 0 & 1 \end{bmatrix} = \begin{bmatrix} -1 & -\frac{5}{4} \\ 0 & 1 \end{bmatrix} \quad (7.2)$$

```
> assign(A=rhs(%));
```

Appliquons la transformation A aux sommets du polygone de la section précédente.

```
> T_Sommets:=A.M_Sommets;
```

$$T_Sommets := \begin{bmatrix} 0 & -4 & -\frac{13}{2} & -7 & -\frac{5}{2} & 0 \\ 0 & 0 & 2 & 4 & 2 & 0 \end{bmatrix} \quad (7.3)$$


Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k),list),k=1..nops
(Sommets_O))];
```

$$Sommets_T := \left[[0,0], [-4,0], \left[-\frac{13}{2}, 2\right], [-7,4], \left[-\frac{5}{2}, 2\right], [0,0] \right] \quad (7.4)$$

Traçons le polygone transformé avec la couleur « turquoise » et superposons-le dans un même graphique avec le polygone original de couleur « rose ».

```
> Polygone_T:=polygon(Sommets_T,color=pink):
display([Polygone_O,Polygone_T],view=[-8..4,-2..4]);
```




Notons que la composition de deux transformations linéaires est une opération qui n'est pas commutative. Cela est évident lorsqu'il y a incompatibilité des dimensions et cela l'est également même s'il y a compatibilité des dimensions dans la composition. Observons-le en commutant les deux transformations précédentes: faisons suivre le cisaillement après la symétrie.

```
> B:=Matrix([ [1,5/4],[0,1] ]).Matrix([ [-1,0],[0,1] ]):
```

```

T_Sommets:=B.M_Sommets:
Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k), list), k=1..nops
(Sommets_O)) ]:
Polygone_T:=polygon(Sommets_T,color=pink):
display([Polygone_O,Polygone_T],scaling=constrained,view=[-8..4,-2..4]);

```



Exemple plus élaboré

L'exemple suivant est un exemple d'une figure qui possède un très grand nombre de points. On réalisera, avec cet exemple, que le souci qu'on a montré précédemment à automatiser l'énumération des coordonnées des sommets transformés est résompensé.

Soit la matrice de rotation $A = \begin{bmatrix} \cos(\theta) & -\sin(\theta) \\ \sin(\theta) & \cos(\theta) \end{bmatrix}$. Choisissons un angle de rotation $\theta = \pi$.

```

> theta:=Pi;
 θ := π |(7.1.1)

```

```

> A:=Matrix([
    [cos(theta), -sin(theta)],
    [sin(theta), cos(theta)] ]);
 A := [ -1  0 ]       [ 0 -1 ] |(7.1.2)

```

Rappelons la procédure que nous avons élaborée dans le document « *Opérandes d'un objet Maple* ». Ce document se retrouve dans l'onglet « *Compléments Maple* » de mon site Internet.


```

> Liste_des_points:=proc(Graphe)
    local Matrice,Nombre_de_points_demandé,Points;
    Matrice:=op(1,op(1,Graphe));
    Nombre_de_points_demandé:=op([1,1],Matrice);
    Points:=[seq([Matrice[k,1],Matrice[k,2]],k=1..Nombre_de_points_demandé)]
    ;
end proc:

```


Tracons une portion de la feuille de cannabis qui a servi d'exemple dans le document « *Opérandes d'un objet Maple* ».

```
> Courbe:=plots[polarplot] ((1+.9*cos(8*t))*(1+.1*cos(24*t))*(1+.1*cos(200*t))*(1+sin(t)), t=-Pi/3..Pi/3, numpoints=1200, axes=None):
Sommets_O:=Liste_des_points(Courbe):
Surface:=plottools[polygon](Sommets_O, color="DarkKhaki"):
display(Surface, thickness=0, view=[-3..3, -3..3]);
```



Exécutons les requêtes suivantes.

```
> M_Sommets:=Matrix([
    [seq(op(1, Sommets_O[i]), i=1..nops(Sommets_O))],
    [seq(op(2, Sommets_O[i]), i=1..nops(Sommets_O))] ]):
T_Sommets:=A.M_Sommets:
Sommets_T:= [seq(convert(SubMatrix(T_Sommets, 1..2, k), list), k=1..nops
(Sommets_O))]:
Polygone_T:=polygon(Sommets_T, color=pink):
display([Surface, Polygone_T], thickness=0, view=[-3..3, -3..3]);
```



Matrices et graphisme par ordinateur

L'écran de l'ordinateur est composé de points (pixels) disposés régulièrement à l'horizontal et à la vertical, chaque point est référencé par le système de coordonnées cartésiennes. Une image peut donc être représentée dans l'ordinateur par une matrice. Les mathématiques et particulièrement les matrices et les calculs matriciels sont alors abondamment employés dans les logiciels de graphisme. Dans le cas où les transformations sont des transformations linéaires, on a vu que la matrice associée à la transformation peut être interprétée comme un opérateur qui transforme les coordonnées d'un point permettant la construction d'un nouveau tracé.

Une transformation de base dans les logiciels de traitement d'images est la translation. Malheureusement, la translation d'un tracé dans le plan est une transformation qui n'est pas une transformation linéaire. Rappelons le résultat suivant: toute transformation linéaire $T: V \rightarrow W$ transforme le vecteur nul de V en vecteur nul de W .

$$T(0_V) = 0_W$$

Par exemple, la translation $T: \mathbb{R}^2 \rightarrow \mathbb{R}^2$ définie par $T\left(\begin{bmatrix} x \\ y \end{bmatrix}\right) = \begin{bmatrix} x + 5 \\ y + 10 \end{bmatrix}$ qui réalise un déplacement

horizontal vers la droite de 5 unités et un déplacement vertical vers le haut de 10 unités est une transformation qui n'est pas une transformation linéaire: l'origine n'étant pas préservée.

En conséquence, toutes transformations découlant d'une ou de plusieurs transformations en rapport avec des objets qui n'incluent pas la préservation de l'origine ne seront pas représentables par une matrice associée. Par exemple, on ne pourra pas obtenir la matrice associée à une rotation d'une image autour d'un point ailleurs qu'à l'origine et donc, il s'avère impossible par la suite d'utiliser la multiplication matricielle impliquant seulement la matrice associée pour calculer les coordonnées des nouveaux points.

Cela ne veut pas dire pour autant qu'il est impossible d'utiliser la multiplication matricielle pour calculer de telles transformations. Notons que chaque point $(x, y) \in \mathbb{R}^2$ peut être identifié par le point $(x, y, 1)$ appartenant au plan d'équation $z = 1$ de $\mathbb{R}^3$. Ce plan étant situé à une hauteur de une unité au-dessus du plan xOy . Nous dirons que le point (x, y) possède les coordonnées homogènes $(x, y, 1)$. Les coordonnées d'un point ainsi formulé ne pourront être traitées comme les composantes d'un vecteur colonne de sorte que l'on ne pourra pas les additionner ni les multiplier par un scalaire mais il sera possible de transformer les coordonnées de ce point par la multiplication matricielle.

Par exemple, la translation définie par

$$\begin{bmatrix} x \\ y \end{bmatrix} \rightarrow \begin{bmatrix} x + h \\ y + k \end{bmatrix}$$

qui n'est pas une transformation linéaire sera reformulée en coordonnées homogènes par

$$(x, y, 1) \rightarrow (x + h, y + k, 1)$$

Cette transformation ainsi reformulée peut maintenant s'exprimer avec une multiplication matricielle.

$$\begin{bmatrix} 1 & 0 & h \\ 0 & 1 & k \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} x \\ y \\ 1 \end{bmatrix} = \begin{bmatrix} x + h \\ y + k \\ 1 \end{bmatrix}$$

De plus, toute transformation linéaire vers $\mathbb{R}^2$ peut être représentée en termes de coordonnées homogènes par une matrice en blocs de la forme

$$\begin{bmatrix} A & 0 \\ 0 & 1 \end{bmatrix}$$

où la matrice A est la matrice associée à la transformation linéaire. Ainsi, toutes les transformations linéaires élémentaires pourront être transposées dans la forme d'une matrice en blocs.

Exemples:

Rotation d'angle θ autour de l'origine est
$$\begin{bmatrix} \cos(\theta) & -\sin(\theta) & 0 \\ \sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Réflexion autour de la droite $y = x$ est
$$\begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$


Étirement selon les deux axes est
$$\begin{bmatrix} k_1 & 0 & 0 \\ 0 & k_2 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Nous allons montrer, avec l'exemple ci-dessous, que la multiplication matricielle de matrices en blocs est également valable pour exprimer la composition de plusieurs transformations.

Soit le triangle suivant.

```
> Sommets_O := [[1,1], [4,1], [3,2], [1,1]]:
```

```
> Triangle_O := polygon(Sommets_O, color=turquoise):  
display(Triangle_O, view=[-1..4, -1..3]);
```



Obtenons **la matrice** en blocs afin de composer les transformation suivantes.

- une homothétie de centre $(\frac{4}{3}, \frac{8}{3})$ et de rapport $\frac{1}{2}$
- suivi d'une rotation

$$- \theta = \frac{\pi}{2}$$

-et finalement, suivi d'une translation de 3 unités vers le haut.

En premier, créons chaque matrice en blocs des transformations requises.

Puisque le centre de l'homothétie n'est pas l'origine, on doit d'abord translater tout le plan à l'origine avant d'effectuer un étirement. Cette translation doit se faire de telle manière que le centre de l'homothétie soit ramené à l'origine du repère.

$$(x, y, 1) \rightarrow (x - \frac{8}{3}, y - \frac{4}{3}, 1)$$

Soit une première matrice en blocs permettant le calcul des coordonnées homogènes de tous les points translats du plan et donc en particulier ceux du triangle. Cela permettra, par la suite, d'opérer des calculs de transformations linéaires élémentaires.

```
> h:=-8/3:
k:=-4/3:
Translation_1:=Matrix([
    [1,0,h],
    [0,1,k],
    [0,0,1]]) ;
```

$$Translation_1 := \begin{bmatrix} 1 & 0 & -\frac{8}{3} \\ 0 & 1 & -\frac{4}{3} \\ 0 & 0 & 1 \end{bmatrix} \quad (8.1)$$

Ensuite, posons la matrice en blocs permettant d'obtenir une homothétie de centre (0,0) et de rapport $\frac{1}{2}$ sur les coordonnées homogènes des points du triangle.

```
> r:=1/2:
Homothetie:=Matrix([
    [r,0,0],
    [0,r,0],
    [0,0,1] ] ) ;
```

$$Homothetie := \begin{bmatrix} \frac{1}{2} & 0 & 0 \\ 0 & \frac{1}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (8.2)$$

Posons maintenant la matrice en blocs permettant d'obtenir une rotation d'angle $\theta = \frac{\pi}{2}$ autour de l'origine des coordonnées homogènes des points du triangle.


```

> theta:=Pi/2:
Rotation:=Matrix([
    [cos(theta), -sin(theta), 0],
    [sin(theta),  cos(theta), 0],
    [          0,          0, 1] ]);

```

$$Rotation := \begin{bmatrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (8.3)$$

Posons la matrice en blocs de translation permettant de ramener tout le plan et donc le centre de l'homothétie à sa position de départ.

```

> h:=8/3:
k:=4/3:
Translation_2:=Matrix([
    [1,0,h],
    [0,1,k],
    [0,0,1]]);

```

$$Translation_2 := \begin{bmatrix} 1 & 0 & \frac{8}{3} \\ 0 & 1 & \frac{4}{3} \\ 0 & 0 & 1 \end{bmatrix} \quad (8.4)$$

Finalement, posons la matrice en blocs permettant un déplacement horizontal de 0 unité et un déplacement vertical de 3 unités tel que demandé.

```

> h:=0:
k:=3:
Translation_3:=Matrix([
    [1,0,h],
    [0,1,k],
    [0,0,1]]);

```

$$Translation_3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 3 \\ 0 & 0 & 1 \end{bmatrix} \quad (8.5)$$

Rendons libres les variables suivantes:

```

[> r:='r':theta:='theta':h:='h':k:='k':

```

Obtenons l'unique matrice en blocs composant l'ensemble des transformations précédentes.

```

> A:=`.`(Translation_3,Translation_2, Rotation,Homothetie,Translation_1);

```

$$A := \begin{bmatrix} 0 & -\frac{1}{2} & \frac{10}{3} \\ \frac{1}{2} & 0 & 3 \\ 0 & 0 & 1 \end{bmatrix} \quad (8.6)$$

Posons la matrice colonnes des « coordonnées homogènes » des sommets du triangle original.

```
> T_Sommets_Homogenes:=Matrix([
    [1,4,3,1],
    [1,1,2,1],
    [1,1,1,1] ]);
```

$$T_Sommets_Homogenes := \begin{bmatrix} 1 & 4 & 3 & 1 \\ 1 & 1 & 2 & 1 \\ 1 & 1 & 1 & 1 \end{bmatrix} \quad (8.7)$$

Obtenons maintenant les images des sommets du triangle original de couleur « truquoise » par la transformation T en calculant le produit matriciel $AM_Sommets$.

```
> Sommets_T_Homogene:=A . T_Sommets_Homogenes;
```

$$Sommets_T_Homogene := \begin{bmatrix} \frac{17}{6} & \frac{17}{6} & \frac{7}{3} & \frac{17}{6} \\ \frac{7}{2} & 5 & \frac{9}{2} & \frac{7}{2} \\ 1 & 1 & 1 & 1 \end{bmatrix} \quad (8.8)$$

Extrayons dans une matrice les coordonnées cartésiennes des points transformés avec la macro-commande SubMatrix.

```
> Sommets_T_Cartesiens:=SubMatrix(Sommets_T_Homogene,1..2,1..4);
```

$$Sommets_T_Cartesiens := \begin{bmatrix} \frac{17}{6} & \frac{17}{6} & \frac{7}{3} & \frac{17}{6} \\ \frac{7}{2} & 5 & \frac{9}{2} & \frac{7}{2} \end{bmatrix} \quad (8.9)$$

Automatisons l'énumération des coordonnées des sommets transformés avec la requête suivante.

```
> Sommets_T:= [seq(convert(SubMatrix(Sommets_T_Cartesiens,1..2,k),list),
    k=1..nops(Sommets_O))];
```

$$Sommets_T := \left[\left[\frac{17}{6}, \frac{7}{2} \right], \left[\frac{17}{6}, 5 \right], \left[\frac{7}{3}, \frac{9}{2} \right], \left[\frac{17}{6}, \frac{7}{2} \right] \right] \quad (8.10)$$

Traçons le triangle transformé avec la couleur « rose » et superposons-le dans un même graphique avec le triangle original de couleur « turquoise ».

```
> Triangle_T:=polygon(Sommets_T,color=pink):
display([Triangle_O,Triangle_T],view=[-1..5,-1..5]);
```

Bingo !

